

6. Folgen und Grenzwerte

6.1 Ermittlung von Grenzwerten

Der Grenzwert einer Zahlenfolge a_n berechnet man in MAPLE mit dem Befehl

```
limit(a(n), n=infinity);
```

```
> a := n -> 1+1/2^n:
```

```
> Limit (a(n), n = infinity) = limit (a(n), n = infinity);
```

$$\lim_{n \rightarrow \infty} 1 + \frac{1}{2^n} = 1$$

Wir stellen den Grenzwert zusammen mit einer ε -Umgebung als Funktions-schaubild für die Folge $(1 + \frac{1}{n})^n$ graphisch dar.

```
> a := n -> (1 + 1/n)^n:
```

```
> folge := [seq([n, a(n)], n = 1..100)]:
```

```
> p1 := plot (folge, style=point):
```

```
> n := 'n': eps := 5 * 10 ^(-2):
```

```
> Grenzwert := limit (a(n), n=infinity);
```

$$\text{Grenzwert} := e$$

```
> p2 := plot ([Grenzwert-eps,Grenzwert,Grenzwert+eps], x = 0..100, 2..3):
```

```
> with (plots): display ([p1, p2]);
```




Abb. 6.1. Grenzwert der Folge $(1 + \frac{1}{n})^n$



Visualisierung mit MAPLE: Auf der Homepage befindet sich ein MAPLE-Worksheet, bei dem man selbst Folgen spezifizieren kann und diese Folgen dann - sofern sie einen Grenzwert besitzen - in Form einer Animation dargestellt werden.

Beispiel 6.1 (Babylonisches Wurzelziehen, mit MAPLE-Worksheet):

Die rekursiv definierte Folge

$$a_0 = a \quad , \quad a_{n+1} = \frac{1}{2} \left(a_n + \frac{a}{a_n} \right)$$

ist für jedes $a > 0$ eine monoton fallende Folge, die gegen $\sqrt{a}$ konvergiert.

> z:=2: #Aus dieser Zahl wird die Wurzel gezogen

> n:=5: #Anzahl der Iterationen

>

> a:=z:

> for i from 1 to n

> do

> a:=1./2*(a+z/a);

> print('Näherung für ',sqrt(z) = a);

> end do:

Näherung für , $\sqrt{2} = 1.500000000$ Näherung für , $\sqrt{2} = 1.416666666$ Näherung für , $\sqrt{2} = 1.414215686$ Näherung für , $\sqrt{2} = 1.414213562$ Näherung für , $\sqrt{2} = 1.414213562$

□

6.2**6.2 Graphische Darstellung von Funktionsfolgen**

In MAPLE kann man Funktionsfolgen anschaulich darstellen, indem sowohl die Folge $(x_n)_n$ als auch die Funktionsfolge $(f(x_n))_n$ in ein Schaubild gezeichnet werden. Zur graphischen Darstellung wählen $f(x) = x^2$ und als Folge $x_n = 2 - \frac{1}{n^2} \xrightarrow{n \rightarrow \infty} 2$. Gesucht ist der Funktionsgrenzwert $\lim_{n \rightarrow \infty} f(x_n)$:

> f := x -> x^2:


> x := n -> 2 - 1/n^2:

> tabelle := n -> [[x(n), 0], [x(n), f(x(n))], [0, f(x(n))]]:

> p1 := plot ([seq(tabelle(i), i = 1..10)], color = blue):

> p2 := plot ([x, f(x), x = 0..2.1], x = 0..2.5, thickness = 2):

> with (plots): display ([p1, p2]);

Abb. 6.2. Linksseitiger Funktionsgrenzwert bei $x_0 = 2$

6.3 Berechnung von Funktionsgrenzwerten

Mit MAPLE lassen sich rechts- und linksseitige Grenzwerte ebenfalls mit dem **limit**-Befehl berechnen, wenn man zusätzlich die Option *left* oder *right* setzt:

Beispiel 6.2. $\lim_{x \rightarrow 0} \frac{\sin(x)}{x}.$

> f(x) := sin(x)/x:

> limit (f(x), x = 0, right);

1

> Limit (f(x), x = 0) = limit (f(x), x = 0);

$$\lim_{x \rightarrow 0} \frac{\sin x}{x} = 1$$

□

Beispiel 6.3. $\lim_{x \rightarrow 0} \frac{e^x - 1}{x}.$

> f(x) := (exp(x) - 1)/x:

> Limit (f(x), x = 0) = limit (f(x), x = 0);

$$\lim_{x \rightarrow 0} \frac{e^x - 1}{x} = 1$$

□

6.4 Bisektionsverfahren

Grundlage für eine einfache numerische Methode zur Bestimmung von Nullstellen einer Funktion bildet der anschauliche Satz: Jede stetige Funktion, die auf einem Intervall $[a, b]$ einen Vorzeichenwechsel hat, besitzt in diesem Intervall eine Nullstelle (siehe Abb. 6.3):




Abb. 6.3. Intervallhalbierungs-Methode

Die Idee des numerischen Algorithmus besteht darin, dass man zu gegebenem Intervall $[a, b]$ die Intervallmitte m bestimmt und die Funktionswerte

$$f(a), f(m), f(b)$$

miteinander vergleicht: Man ersetzt den Intervallrand durch m , dessen Funktionswert dasselbe Vorzeichen wie $f(m)$ besitzt. Anschließend wiederholt man

die Vorgehensweise auf dem halbierten Intervall usw. Da man ständig die Intervalllänge halbiert, nennt man das Verfahren auch Intervallhalbierungs-Methode.

Umsetzung mit MAPLE. Bei der Realisierung der Intervallhalbierungs-Methode mit MAPLE wird der Algorithmus direkt übernommen.

```
> bise := proc()
> local iter, x1, x2, x3, f1, f2, f3, delta, f, func, x;
>
> func := args[1]: x := op(1, args[2]);
> f := unapply (func, x):
> x1 := op(1, op(2, args[2]));
> x2 := op(2, op(2, args[2]));
> f1 := f(x1): f2 := f(x2):
>
> iter := 0: delta := 1e-4:
> while x2 - x1 > delta
> do iter := iter + 1:
>   x3 := (x2 + x1)/2.: f3 := f(x3):
>   if (evalf (f3 * f2) <= 0) then x1 := x3: f1 := f3:
>   else x2 := x3: f2 := f3:
>   fi;
> lprint ('[', x1, ', ', x2, ']'):
> end do;
> print ('Die Nullstelle liegt nach ', iter, 'Iterationen bei xi = ', x3);
> end;
```

Der Aufruf der Prozedur **bise** erfolgt wie der **plot**-Aufruf für einen Ausdruck.

```
> bise (x^3 - sqrt(x^2 + 1), x = 1..2);
```

[	1.	,	1.500000000	]
[	1.	,	1.250000000	]
[	1.125000000	,	1.250000000	]
[	1.125000000	,	1.187500000	]
[	1.125000000	,	1.156250000	]
[	1.140625000	,	1.156250000	]
[	1.148437500	,	1.156250000	]
[	1.148437500	,	1.152343750	]
[	1.150390625	,	1.152343750	]
[	1.150390625	,	1.151367188	]
[	1.150878907	,	1.151367188	]
[	1.150878907	,	1.151123048	]
[	1.150878907	,	1.151000978	]

Die Nullstelle liegt nach , 13, Iterationen bei xi = , 1.150939943



Visualisierung mit MAPLE: Auf der Homepage befindet sich eine erweiterte MAPLE-Prozedur, **bise_ext**, die den Konvergenzprozess in Form einer Animation visualisiert. Darüber hinaus gibt es auf der Homepage ein eigenes Kapitel über das numerische Lösen von Gleichungen.

6.5 Zusammenstellung der MAPLE-Befehle

6.5

Grenzwertbildung mit MAPLE

$a := n \rightarrow 1/n$	Definition der Folge $a_n = \frac{1}{n}$.
limit ($a(n)$, $n=infinity$)	Berechnung des Grenzwertes $\lim_{n \rightarrow \infty} a_n$.
Limit ($a(n)$, $n=infinity$)	Symbolische Darstellung des Grenzwertes.
limit ($f(x)$, $x=x0$)	Berechnung des Grenzwertes $\lim_{x \rightarrow x_0} f(x)$.
Limit ($f(x)$, $x=x0$)	Symbolische Darstellung des Grenzwertes.

MAPLE-Worksheets zu Kapitel 6



Die folgenden elektronischen Arbeitsblätter stehen für Kapitel 6 mit MAPLE zur Verfügung.

- Zahlenfolgen
- Babylonisches Wurzelziehen
- Funktionsfolgen
- Bisektionsmethode



<http://www.springer.com/978-3-642-25052-1>

Ingenieurmathematik kompakt mit Maple

Westermann, Th.

2012, X, 252 S. 100 Abb., Softcover

ISBN: 978-3-642-25052-1